# NAG Library Routine Document

# D05AAF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

D05AAF solves a linear, nonsingular Fredholm equation of the second kind with a split kernel.

## 2 Specification

```
 SUBROUTINE D05AAF(LAMBDA, A, B, K1, K2, G, F, C, N, IND, W1, W2, WD,
1                  LDW1, LDW2, IFAIL)
 INTEGER           N, IND, LDW1, LDW2, IFAIL
 double precision  LAMBDA, A, B, K1, K2, G, F(N), C(N), W1(LDW1,LDW2),
1                  W2(LDW2,4), WD(LDW2)
 EXTERNAL          K1, K2, G
```

## 3 Description

D05AAF solves an integral equation of the form

$$f(x) - \lambda \int_a^b k(x,s) f(s)\, ds = g(x)$$

for $a \leq x \leq b$, when the kernel $k$ is defined in two parts: $k = k_1$ for $a \leq s \leq x$ and $k = k_2$ for $x < s \leq b$. The method used is that of El–Gendi (1969) for which, it is important to note, each of the functions $k_1$ and $k_2$ must be defined, smooth and nonsingular, for all $x$ and $s$ in the interval $[a, b]$.

An approximation to the solution $f(x)$ is found in the form of an $n$ term Chebyshev series $\sum_{i=1}^{n}{}' c_i T_i(x)$, where $'$ indicates that the first term is halved in the sum. The coefficients $c_i$, for $i = 1, 2, \ldots, n$, of this series are determined directly from approximate values $f_i$, for $i = 1, 2, \ldots, n$, of the function $f(x)$ at the first $n$ of a set of $m + 1$ Chebyshev points:

$$x_i = \tfrac{1}{2}(a + b + (b - a)\cos[(i-1)\pi/m]), \qquad i = 1, 2, \ldots, m+1.$$

The values $f_i$ are obtained by solving simultaneous linear algebraic equations formed by applying a quadrature formula (equivalent to the scheme of Clenshaw and Curtis (1960)) to the integral equation at the above points.

In general $m = n - 1$. However, if the kernel $k$ is centro-symmetric in the interval $[a, b]$, i.e., if $k(x,s) = k(a + b - x, a + b - s)$, then the routine is designed to take advantage of this fact in the formation and solution of the algebraic equations. In this case, symmetry in the function $g(x)$ implies symmetry in the function $f(x)$. In particular, if $g(x)$ is even about the mid-point of the range of integration, then so also is $f(x)$, which may be approximated by an even Chebyshev series with $m = 2n - 1$. Similarly, if $g(x)$ is odd about the mid-point then $f(x)$ may be approximated by an odd series with $m = 2n$.

## 4 References

Clenshaw C W and Curtis A R (1960) A method for numerical integration on an automatic computer *Numer. Math.* **2** 197–205

El–Gendi S E (1969) Chebyshev solution of differential, integral and integro-differential equations *Comput. J.* **12** 282–287

## 5 Parameters

1: LAMBDA – ***double precision*** *Input*

*On entry*: the value of the parameter $\lambda$ of the integral equation.

2: A – ***double precision*** *Input*

*On entry*: $a$, the lower limit of integration.

3: B – ***double precision*** *Input*

*On entry*: $b$, the upper limit of integration.

*Constraint*: B > A.

4: K1 – ***double precision*** FUNCTION, supplied by the user. *External Procedure*

K1 must evaluate the kernel $k(x,s) = k_1(x,s)$ of the integral equation for $a \leq s \leq x$.

The specification of K1 is:

```
double precision FUNCTION K1(X, S)
double precision             X, S
```

1: X – ***double precision*** *Input*
2: S – ***double precision*** *Input*

*On entry*: the values of $x$ and $s$ at which $k_1(x,s)$ is to be evaluated.

K1 must be declared as EXTERNAL in the (sub)program from which D05AAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

5: K2 – ***double precision*** FUNCTION, supplied by the user. *External Procedure*

K2 must evaluate the kernel $k(x,s) = k_2(x,s)$ of the integral equation for $x < s \leq b$.

The specification of K2 is:

```
double precision FUNCTION K2(X, S)
double precision             X, S
```

1: X – ***double precision*** *Input*
2: S – ***double precision*** *Input*

*On entry*: the values of $x$ and $s$ at which $k_2(x,s)$ is to be evaluated.

K2 must be declared as EXTERNAL in the (sub)program from which D05AAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

Note that the functions $k_1$ and $k_2$ must be defined, smooth and nonsingular for all $x$ and $s$ in the interval $[a,b]$.

6: G – ***double precision*** FUNCTION, supplied by the user. *External Procedure*

G must evaluate the function $g(x)$ for $a \leq x \leq b$.

The specification of G is:

```
double precision FUNCTION G(X)
double precision            X
```

1: X – ***double precision*** *Input*

*On entry*: the values of $x$ at which $g(x)$ is to be evaluated.

G must be declared as EXTERNAL in the (sub)program from which D05AAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

7: F(N) – ***double precision*** array *Output*

*On exit*: the approximate values $f_i$, for $i = 1, 2, \ldots, \mathrm{N}$, of $f(x)$ evaluated at the first N of $\mathrm{M} + 1$ Chebyshev points $x_i$, (see Section 3).

If $\mathrm{IND} = 0$ or 3, $\mathrm{M} = \mathrm{N} - 1$.

If $\mathrm{IND} = 1$, $\mathrm{M} = 2 \times \mathrm{N}$.

If $\mathrm{IND} = 2$, $\mathrm{M} = 2 \times \mathrm{N} - 1$.

8: C(N) – ***double precision*** array *Output*

*On exit*: the coefficients $c_i$, for $i = 1, 2, \ldots, \mathrm{N}$, of the Chebyshev series approximation to $f(x)$.

If $\mathrm{IND} = 1$ this series contains polynomials of odd order only and if $\mathrm{IND} = 2$ the series contains even order polynomials only.

9: N – INTEGER *Input*

*On entry*: the number of terms in the Chebyshev series required to approximate $f(x)$.

*Constraint*: $\mathrm{N} > 0$.

10: IND – INTEGER *Input*

*On entry*: determines the forms of the kernal, $k(x, s)$, and the function $g(x)$.

$\mathrm{IND} = 0$
: $k(x, s)$ is not centro-symmetric (or no account is to be taken of centro-symmetry).

$\mathrm{IND} = 1$
: $k(x, s)$ is centro-symmetric and $g(x)$ is odd.

$\mathrm{IND} = 2$
: $k(x, s)$ is centro-symmetric and $g(x)$ is even.

$\mathrm{IND} = 3$
: $k(x, s)$ is centro-symmetric but $g(x)$ is neither odd nor even.

*Constraint*: $\mathrm{IND} = 0$, 1, 2 or 3.

11: W1(LDW1,LDW2) – ***double precision*** array *Workspace*
12: W2(LDW2,4) – ***double precision*** array *Workspace*
13: WD(LDW2) – ***double precision*** array *Workspace*

14: LDW1 – INTEGER *Input*

*On entry*: the first dimension of the array W1 as declared in the (sub)program from which D05AAF is called.

*Constraint*: $\mathrm{LDW1} \geq \mathrm{N}$.

15: LDW2 – INTEGER *Input*

*On entry*: the second dimension of the array W1 and the first dimension of the array W2 and the dimension of the array WD as declared in the (sub)program from which D05AAF is called.

*Constraint*: $\mathrm{LDW2} \geq 2 \times \mathrm{N} + 2$.

16: IFAIL – INTEGER *Input/Output*

*On entry*: IFAIL must be set to 0, $-1$ or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

*On exit*: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value $-1$ or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value $-1$ or 1 is used it is essential to test the value of IFAIL on exit.**

# 6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or $-1$, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, $\mathrm{A} \geq \mathrm{B}$ or $\mathrm{N} < 1$.

IFAIL $= 2$

A failure has occurred due to proximity to an eigenvalue. In general, if LAMBDA is near an eigenvalue of the integral equation, the corresponding matrix will be nearly singular. In the special case, $m = 1$, the matrix reduces to a zero-valued number.

# 7 Accuracy

No explicit error estimate is provided by the routine but it is usually possible to obtain a good indication of the accuracy of the solution either

(i) by examining the size of the later Chebyshev coefficients $c_i$, or

(ii) by comparing the coefficients $c_i$ or the function values $f_i$ for two or more values of N.

# 8 Further Comments

The time taken by D05AAF increases with N.

This routine may be used to solve an equation with a continuous kernel by defining K1 and K2 to be identical.

This routine may also be used to solve a Volterra equation by defining K2 (or K1) to be identically zero.

# 9 Example

This example solves the equation

$$f(x) - \int_0^1 k(x,s) f(s)\, ds = \left(1 - \frac{1}{\pi^2}\right) \sin(\pi x)$$

where

$$k(x,s) = \begin{cases} s(1-x) & \text{for } 0 \leq s < x, \\ x(1-s) & \text{for } x \leq s \leq 1. \end{cases}$$

Five terms of the Chebyshev series are sought, taking advantage of the centro-symmetry of the $k(x,s)$ and even nature of $g(x)$ about the mid-point of the range $[0,1]$.

The approximate solution at the point $x = 0.1$ is calculated by calling C06DBF.

## 9.1 Program Text

```
*     D05AAF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          N, LDW1, LDW2
      PARAMETER        (N=5,LDW1=N,LDW2=2*N+2)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Scalars in Common ..
      DOUBLE PRECISION R
*     .. Local Scalars ..
      DOUBLE PRECISION A, ANS, B, LAMBDA, X
      INTEGER          I, IFAIL, IND, IS
*     .. Local Arrays ..
      DOUBLE PRECISION C(LDW1), F(LDW1), W1(LDW1,LDW2), W2(LDW2,4),
     +                 WD(LDW2)
*     .. External Functions ..
      DOUBLE PRECISION C06DBF, G, K1, K2, X01AAF
      EXTERNAL         C06DBF, G, K1, K2, X01AAF
*     .. External Subroutines ..
      EXTERNAL         D05AAF
*     .. Common blocks ..
      COMMON           R
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D05AAF Example Program Results'
      WRITE (NOUT,*)
      R = X01AAF(0.0D0)
      LAMBDA = 1.0D0
      A = 0.0D0
      B = 1.0D0
      IND = 2
      IFAIL = 1
*
      CALL D05AAF(LAMBDA,A,B,K1,K2,G,F,C,N,IND,W1,W2,WD,LDW1,LDW2,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,*)
     +'Kernel is centro-symmetric and G is even so the solution is even'
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Chebyshev coefficients'
         WRITE (NOUT,*)
         WRITE (NOUT,99998) (C(I),I=1,N)
         WRITE (NOUT,*)
         X = 0.1D0
*        Note that X has to be transformed to range [-1,1]
         IS = 1
         IF (IND.EQ.1) THEN
            IS = 3
         ELSE
            IF (IND.EQ.2) IS = 2
         END IF
         ANS = C06DBF(2.0D0/(B-A)*(X-0.5D0*(B+A)),C,N,IS)
         WRITE (NOUT,99999) 'X=', X, '    ANS=', ANS
      ELSE
         WRITE (NOUT,99997) IFAIL
      END IF
*
99999 FORMAT (1X,A,F5.2,A,1F10.4)
99998 FORMAT (1X,5E14.4)
99997 FORMAT (1X,' ** D05DADF returned with IFAIL = ',I5)
      END
*
      DOUBLE PRECISION FUNCTION K1(X,S)
*     .. Scalar Arguments ..
      DOUBLE PRECISION S, X
*     .. Executable Statements ..
      K1 = S*(1.0D0-X)
      RETURN
      END
*
```

```
      DOUBLE PRECISION FUNCTION K2(X,S)
*     .. Scalar Arguments ..
      DOUBLE PRECISION S, X
*     .. Executable Statements ..
      K2 = X*(1.0D0-S)
      RETURN
      END
*
      DOUBLE PRECISION FUNCTION G(X)
*     .. Scalar Arguments ..
      DOUBLE PRECISION X
*     .. Scalars in Common ..
      DOUBLE PRECISION R
*     .. Intrinsic Functions ..
      INTRINSIC        SIN
*     .. Common blocks ..
      COMMON           R
*     .. Executable Statements ..
      G = SIN(R*X)*(1.0D0-1.0D0/(R*R))
      RETURN
      END
```

## 9.2 Program Data

None.

## 9.3 Program Results

```
 D05AAF Example Program Results

 Kernel is centro-symmetric and G is even so the solution is even

 Chebyshev coefficients

     0.9440E+00   -0.4994E+00    0.2799E-01   -0.5967E-03    0.6658E-05

 X= 0.10    ANS=    0.3090
```